
Abstract
Symbolic execution is a technique for automatic software
validation and verification. New symbolic executors regu-
larly appear for both existing and new languages and such
symbolic executors are generally manually (re)implemented
each time we want to support a new language. We propose
to automatically generate symbolic executors from language
definitions, and present a technique for mechanically (but
as yet, manually) deriving a symbolic executor from a defini-
tional interpreter. The idea is that language designers define
their language as a monadic definitional interpreter, where
the monad of the interpreter defines the meaning of branch
points. Developing a symbolic executor for a language is
a matter of changing the monadic interpretation of branch
points. In this paper, we illustrate the technique on a lan-
guage with recursive functions and pattern matching, and
use the derived symbolic executor to automatically generate
test cases for definitional interpreters implemented in our
defined language.

CCS Concepts • Theory of computation → Program
schemes; • Software and its engineering→ Formal meth-
ods; Automatic programming.

Keywords Symbolic Execution, Monads, Haskell, Defini-
tional Interpreter

1 Introduction
Symbolic execution [27] is a meta-programming technique
that is at the core of techniques for boosting developer pro-
ductivity, such as automated testing [3, 9, 17, 19, 38] and
program synthesis [14, 20, 35]. A symbolic executor allows
exploration of possible execution paths by running a pro-
gram with symbolic variables in place of concrete values.
By strategically instantiating symbolic variables, a symbolic
executor can be used to systematically analyze which parts
of a program are reachable, with which inputs.
Constructing symbolic executors is non-trivial, and en-

abling support for symbolic execution for general-purpose
languages, such as C [4, 19, 38], C++ [31], Java [1, 37], PHP [2],
or Rust [33], is the topic of entire publications at major soft-
ware engineering conferences. We propose that techniques
for symbolic execution are reusable between languages, and
investigate the foundations of how to define and implement
symbolic executors, by deriving them from definitional inter-
preters. Our long-term goal is to integrate these techniques
into languageworkbenches, such as Spoofax [25], Rascal [29],
or Racket [15], to enable the automatic generation of pro-
grammer productivity boosting tools, such as automated
testing frameworks and program synthesizers.

In this paper we explore how to mechanically derive sym-
bolic executors that explore possible execution paths through
programs by instantiating and specializing symbolic vari-
ables, following a breadth-first search strategy. Our explo-
ration revolves around a dynamically-typed language with
recursive functions and pattern matching. Using Haskell as
our meta-language, and working with its integrated sup-
port for generic and monadic programming, we implement
a definitional interpreter for this language. This definitional
interpreter is parameterized with an interface which we in-
stantiate in two different ways to obtain first a concrete
interpreter, and then a symbolic executor. The “derivation”
thus amounts to instantiating the interface operations in a
manner that yields a symbolic executor.
The symbolic executor we derive allows us to explore

the solution space for constraints such as the following con-
straint that a list xs must be a palindrome:

xs ≡ reverse xs

Symbolic execution explores all execution paths through the
reverse function that satisfy the constraint, and instantiates
xs accordingly, thereby generating palindromes. This paper
is a literate Haskell file, and we invite interested readers to
download the Haskell version of the paper to experiment
with, and extend, the framework we present.1

Related Previous Lines of Work The techniques that we
develop in this paper are closely related to the techniques
used for relational programming, pioneered by Friedman
and Byrd in miniKanren [5, 8, 16, 22], a language for re-
lational programming and constraint logic programming,
which has been implemented in a wide range of different
languages; notably Scheme [7, 16], but also, e.g., OCaml [30].
The miniKanren language and many of its implementations
have been developed and researched for more than a decade,
with new developments and improvements appearing each
year, such as new and better heuristics for guiding the ex-
ploration of execution paths [34]. The motivation for this
paper is to bring similar benefits as found in miniKanren to
programming languages at large, by automatically deriving
symbolic executors from definitional interpreters.
Rosette [40, 41] is a solver-aided language that extends

Racket [15] to provide framework for implementing solver-
aided domain-specific languages, by means of a symbolic
virtual machine and symbolic compiler. This VM brings the
benefits of symbolic execution and model checking to lan-
guages implemented in Rosette via general-purpose symbolic
abstractions that support sophisticated symbolic reasoning,
beyond the relatively simple constraints found in (most vari-
ants of) miniKanren. A main goal of Rosette is to implement
solver-aided languages, but the symbolic abstractions and
techniques that Rosette implements could also be used to
address the problem that is the motivation for this paper,
namely the problem of automatically deriving symbolic ex-
ecutors from “traditional” definitional interpreters.

There has been much work on symbolic execution in the
literature on software engineering; e.g., [1, 2, 4, 19, 31, 33, 37,
38]. Many of these frameworks are so-called concolic frame-
works that work by instrumenting a concrete language run-
time to track symbolic path constraints. After each concrete
execution, these path constraints are collected and solved
in order to cover a different path through the program in a
subsequent run of the program. Concolic testing is typically
implemented by generating test inputs randomly, rather than
systematically solving path constraints. In this paper, we ex-
plore a symbolic execution strategy which interleavingly
explores multiple execution paths concurrently, rather than
a concolic testing approach which would require a relatively
sophisticated constraint solver in order to explore execution
paths in an equally systematic manner.

1https://github.com/MetaBorgCube/From-Definitional-Interpreter-To-
Symbolic-Executor

Contributions

• Techniques (in § 3) for deriving symbolic executors
from definitional interpreters, by using free monads to
compile programs into command trees, and interpret-
ing these using a small-step execution strategy.
• A symbolic executor (in § 4) for a language with alge-
braic datatypes that illustrates these techniques.
• A simple example application (in § 6): automated test
generation for definitional interpreters.

The rest of this paper is structured as follows. In § 2 we
introduce a definitional interpreter for a language with recur-
sion and pattern matching. In § 3 we present a definitional
interpretation of the effects, by means of a free monad, using
a small-step semantics execution strategy. In § 4 we gener-
alize the definitional interpretation of effects from § 3, to
obtain a symbolic executor, whose correctness we discuss
in § 5. Finally, in § 6 we discuss a case study application
of the symbolic executor: generating tests for definitional
interpreters, and § 7 concludes.

2 Definitional Interpreter for a Language
With Pattern Matching

Definitional interpreters define the meaning of a (new) ob-
ject language by implementing an interpreter for it in an
existing, well-understood, language. We use Haskell to im-
plement a definitional interpreter for a functional language
with pattern matching.

2.1 Syntax
The abstract syntax of the language we consider is summa-
rized in Fig. 1. The expression constructors for Var , Lam, and
App are standard expressions for variables, unary functions,
and function application. An expression constructor expres-
sion Con f [e1, ..., en ] represents an n-ary term whose head
symbol is f , andwhose sub-term values are the results of eval-
uating each expression e1 ...en. Case e [(p1, e1), ..., (pn, en)] is
a pattern match expression which first evaluates e to a value
and then attempts to match the resulting value against the
patterns p1 ... pn, where patterns are given by the type Patt.
Letrec expressions are restricted to bind value expressions,
given by the type ValExpr .

2.2 Prelude to a Definitional Interpreter:
Effects and Values

The definitional interpreter for the language we consider
in this paper is given in Fig. 2. The interpreter depends on
the EffVal type class which in turn depends on a number
of type classes that constrain the polymorphic notion of
effects (defined by a monad m) and values (defined by a
value type val) of the interpreter. The EffVal type class is
thus a polymorphic embedding [23] of a language that allows
us to define a family of interpreters for the same language,

data Expr = Con String [Expr ]
| Case Expr [(Patt, Expr)]
| Var String
| Lam String Expr
| App Expr Expr
| Let [(String, Expr)] Expr
| Letrec [(String,ValExpr)] Expr
| EEq Expr Expr

data ValExpr = VCon String [ValExpr ]
| VLam String Expr

data Patt = PVar String
| PCon String [Patt ]

Figure 1. Syntax for a language with pattern matching, func-
tions, let, and letrec

akin to the finally tagless approach of Carette et al. [10]. We
summarize the type classes that EffVal comprises.

Effects The language that we define has two classes of ef-
fects: lexically-scoped functions and pattern matching. The
following Haskell type class constrains a monadm to provide
two operations for accessing environments (ask), and alter-
ing which local environment is passed down to recursive
calls of the interpreter (local):

type Env val = [(String, val)]
class Monad m⇒ MonadEnv val m where
ask ::m (Env val)
local :: (Env val → Env val) → m val → m val

MonadEnv is a specialized version of the classical reader
monad [18, 24, 32]:

class Monad m⇒ ClassicalMonadReader r m where
askc ::m r
localc :: (r → r) → m a→ m a

There are two reasons why we use a specialized version. The
reason we specialize the type of environments, as opposed
to an arbitrary type r , is to help Haskell’s type class instance
resolution engine (using GHC v8.6.4). The reason we insist
that the return type is val for the computation that local
takes as argument, is a desire to know that this particular
computation is value-producing, for reasons we explain § 3.
The purpose of symbolic execution is to decide which

inputs cause which parts of a program to execute. For this
reason, we treat conditional branching as an effect. The fol-
lowing type class constrains a monad m to provide a generic
operation for branching:

class Monad m⇒ MonadBranch cval rval fork m where
branch :: cval → fork m rval → m rval

This type class is parameterized by: (1) a value type cval
that branch selection is conditional upon; (2) a value type
rval for the return type of computations in branches; and
(3) a fork type, an abstract notion of branches comprising
computations described by m and val. To illustrate, consider
the following instance of MonadBranch which represents a
classical if-then-else expression:

newtype IfThenElse m a = ITE (m a,m a)

instance Monad m⇒
MonadBranch Bool rval IfThenElse m where

branch True (ITE (t, )) = t
branch False (ITE ( , f )) = f

For our interpreter, which branches on values and returns
values of the same type, we rely on the following more re-
strictive version of MonadBranch:2

class Monad m⇒ MonadMatch val fork m where
match :: val → fork m val → m val

And our interpreter uses the following notion of fork over
a list of pairs consisting of a pattern and a (monadic) com-
putation where each computation has the same return type
a:

newtype Cases m a = Cases [(Patt,m a)]

Values The following type classes define the constructors
for term values conv and function closures closv , as well as
operation app for applying a function to an argument and
operation eq for checking equality between two term values.

class TermVal val where
conv :: String → [val ] → val

class FunVal val where
closv :: String → Expr → Env val → val

class FunApp val m where
app :: val → val → m val

class TermEq val m where
eq :: val → val → m val

2.3 A Definitional Interpreter for a Language with
Pattern Matching

The interpreter in Fig. 2 relies on the effect and value type
classes summarized in the previous section. Additionally, the
interpreter makes use of a few auxiliary functions whose
definitions we elide: mmap maps a monadic function over a
list; mapSnd maps a function over the second element of a
tuple; and resolve resolves a name in an association list, or
fails. The implementation of Letrec uses Haskell’s support for

2The main motivation for using the more specific notion of MonadMatch
here is to help Haskell’s type class resolution engine (using GHC v8.6.4).
Morally, MonadBranch should do.

interp :: EffVal m val ⇒ Expr → m val

interp (Con c es) = do
vs← mmap interp es
return (conv c vs)

interp (Case e bs) =
let vbs = map (mapSnd interp) bs in do
v ← interp e
match v (Cases vbs)

interp (Var x) = do
nv ← ask
return (resolve x nv)

interp (Lam x e) = do
nv ← ask
return (closv x e nv)

interp (App e1 e2) = do
f ← interp e1
a← interp e2
app f a

interp (Let xes e) = do
nv ← mmap interpSnd xes
local (λnv0 → nv ++ nv0) (interp e)
where interpSnd (x, e) = do
v ← interp e; return (x, v)

interp (Letrec xves e) = do
nv ← ask
let nvb = map (mapSnd (interpval nvr )) xves

nvr = nvb ++ nv in
local (λ → nvr ) (interp e)

interp (EEq e1 e2) = do
v1 ← interp e1
v2 ← interp e2
eq v1 v2

interpval :: (TermVal val, FunVal val) ⇒
Env val → ValExpr → val

interpval nv (VLam x e) = closv x e nv
interpval nv (VCon x es) =
conv x (map (interpval nv) es)

Figure 2. A definitional interpreter for a language with pattern matching

(lazy) recursive definitions to define a recursive environment
nvr that ValExprs are evaluated under.
To run our definitional interpreter we must provide con-

crete instances of the abstract type classes from § 2.2. We
use the following notion of value and monad:

data ConcreteValue = ConV String [ConcreteValue ]
| ClosV String Expr (Env ConcreteValue)

type ConcreteMonad =
ReaderT (Env ConcreteValue) (Except String)

Here ReaderT is a monad transformer [32] for the classi-
cal reader monad, and Except is the exception monad. So
ConcreteMonad is isomorphic to:

type ConcreteMonad ′ a =
Env ConcreteValue→ Either String a

The type class instances for this notion of value and monad
are defined in the obvious way. MonadMatch attempts to
pattern match a value against a list of cases by attempting
each from left-to-right until a match succeeds:

instance MonadMatch ConcreteValue Cases
ConcreteMonad where

match v (Cases ((p,m) : bs)) = case vmatch (v, p) of
Just nv → local (λnv0 → nv ++ nv0) m
Nothing → match v (Cases bs)

match (Cases [ ]) = throwError "Match failure"

vmatch :: (ConcreteValue, Patt) →
Maybe (Env ConcreteValue)

Using these type class instances, our definitional interpreter
can be run as follows:

runSteps :: Expr → Env ConcreteValue→
Either String ConcreteValue

runSteps e nv = runExcept (runReaderT (interp e) nv)

3 Towards a Symbolic Executor
The definitional interpreter presented in § 2.3 uses stan-
dard monads and monad transformers to implement the
definitional interpreter given in Fig. 2. But it gives meta-
programmers little control over how interpretation proceeds.
Our goal is to implement a symbolic executor for running a
program in a way that interleavingly explores all possible ex-
ecution paths. To this end, we want a symbolic executor that
can operate on a pool of concurrently running threads where
each thread represents a possible path through the program.
We will approach this challenge by adopting a small-step
execution strategy for each thread. In this section we provide
alternative type class instances that give meta-programmers
more fine-grained control over how interpretation proceeds.
Concretely, we adopt a small-step execution strategy for
effect interpretation, by using free monads.
Following Kiselyov and Ishii [28] and Swierstra and Baa-

nen [39], the following data type defines a family of free
monads:

data Free c a = Stop a
| ∀b.Step (c b) (b→ Free c a)

14

Following Hancock and Setzer [21], we call values of this
data type command trees: each Step represents an application
of a command c b, corresponding to a monadic operation,
which yields a value of type b when interpreted. This value
is passed to the continuation (b→ Free c a) of Step. The Free
data type is a monad:

instance Monad (Free c) where
return = Stop

Stop a >>= k = k a
Step c f >>= k = Step c (λx → f x >>= k)

By defining a suitable notion of command, we can define a
free monad instance which satisfies the type class constraints
for our definitional interpreter from Fig. 2. The following
data type defines such a notion of command:

data Cmd val :: ∗ → ∗where
Match :: val → Cases (Free (Cmd val)) val →

Cmd val val
Local :: (Env val → Env val) → Free (Cmd val) val →

Cmd val val
Ask :: Cmd val (Env val)
Appc :: val → val → Cmd val val
Eqc :: val → val → Cmd val val
Fail :: String → Cmd val a

By instantiating each of the type classes we obtain a compiler
from expressions into command trees:

comp :: (TermVal val, FunVal val) ⇒
Expr → Free (Cmd val) val

comp = interp

The command trees that comp yields are the sequences (or
rather trees) of effectful operations that define the meaning
of object language expressions. But the meaning of command
trees is left open to interpretation. We define the meaning of
command trees by means of a small-step transition function
and a driver loop for the transition function. This small-
step transition function operates on a single command tree
(whose type we abbreviate Threadc , since the command tree
represents a thread of interpretation), and yields a single
command tree as result (or raises an exception). For brevity,
we show just a few cases of the step function:

type Threadc = Free (Cmd ConcreteValue)

step ::Threadc ConcreteValue→
ConcreteMonad (Threadc ConcreteValue)

step (Stop x) = return (Stop x)
step (Step (Match (Cases [ ])) ) =

throwError "Pattern match failure"

step (Step (Match v (Cases ((p,m) : bs))) k) =
case vmatch (v, p) of
Just nv →

return (Step (Local (λnv0 → nv ++ nv0) m) k)
Nothing → step (Step (Match v (Cases bs)) k)

The driver loop for the step function is straightforwardly
defined to continue interpretation until the current thread
of interpretation terminates successfully (or fails):

drive ::Threadc ConcreteValue→
ConcreteMonad ConcreteValue

drive (Stop x) = return x
drive c = do r ← step c; drive r

Thus an alternative definitional interpreter for the language
in Fig. 2 is given by the following function:

runSteps :: Expr → Env ConcreteValue→
Either String ConcreteValue

runSteps e nv = runExcept (runReaderT (drive (comp e)) nv)

4 From Definitional Interpreter to
Symbolic Executor

In this section we derive a symbolic executor from the defi-
nitional interpreter in § 3, by: (1) generalizing the notion of
value from previous sections to also incorporate symbolic
variables; and (2) generalizing the semantics (monad and
small-step transition function) to support instantiation of
symbolic variables and fork new threads of interpretation.

Symbolic Values The updated notion of value is an exten-
sion of the notion of ConcreteValue data type from § 2.3 with
a symbolic variable constructor, SymV :

data SymbolicValue = ConV ′ String [SymbolicValue ]
| ClosV ′ String Expr

(Env SymbolicValue)
| SymV String

Monad The monad for evaluating a step of symbolic ex-
ecution has an environment and may raise an exception,
just like the monad in § 3 for evaluating a step of concrete
execution. Additionally, the monad has a stateful Int field for
keeping track of a fresh supply of symbolic variable names:

type SymbolicMonad =
ReaderT (Env SymbolicValue)

(StateT Int (Except String))

Since symbolic execution should explore all possible execu-
tion paths through a program, we generalize the small-step
transition relation from § 3 by letting the transition relation
take a single thread of interpretation as input, but return a
set of possible continuation threads. Each step may result in
unifying a symbolic variable in order to explore a possible
execution path. Our generalized notion of monad is thus
given by the following types:

type Unifier = [(String, SymbolicValue)]
type UnifierN = [(SymbolicValue, SymbolicValue)]

type SymbolicSetMonad =
StateT (Unifier,UnifierN ) (ListT SymbolicMonad)

Here, Unifier witnesses how symbolic variables must be in-
stantiated in order to complete a single transition step, rep-
resenting a particular execution path of the program being
symbolically executed. UnifierN represents a set of negative
unification constraints. We motivate the use and need for
these shortly. The ListT monad transformer generalizes the
return type of a monadic computation m a to return a list
of as; i.e., m [a]. Note that, although we call ListT a monad
transformer, it is well-known that ListT in Haskell is not
guaranteed to yield a monad that satisfies the monad laws.
For the purpose of this paper, it is not essential whether the
particular definition of SymbolicSetMonad above actually sat-
isfies the monad laws.

Small-Step Transition Function Our symbolic executor
is derived from the concrete semantics of effects in § 3 by
altering how we Match and Eqc effects are interpreted. Thus
all cases of the transition function steps (below) are identical
to the small-step transition function from § 3, except for the
cases for the Match and Eqc . Furthermore, the definitional
interpreter from Fig. 2 is unchanged. We summarize the in-
teresting cases for the steps function, which takes a symbolic
interpretation thread,Threads , as input, and returns a set of
threads (note the use of SymbolicSetMonad):

type Threads = Free (Cmd SymbolicValue)

steps ::Threads SymbolicValue→
SymbolicSetMonad (Threads SymbolicValue)

steps (Step (Match (Cases [ ])) ) = mzero
steps (Step (Match v (Cases ((p,m) : bs))) k) = (do
(nv, u) ← vmatchs (v, p)
(applySubst u (Step (Local (λnv0 → nv ++ nv0) m) k))

‘mplus‘ steps (Step (Match v (Cases bs)) k))
‘catchError‘ (λ → steps (Step (Match v (Cases bs)) k))

steps (Step (Eqc v1 v2) k) =
case unify v1 v2 of
Just [ ] → return (k (ConV ′ "true" [ ]))
Just u → do
(applySubst u (k (ConV ′ "true" [ ]))) ‘mplus‘
(constrainUnifN u (k (ConV ′ "false" [ ])))

Nothing →
return (k (ConV ′ "false" [ ]))

As in § 3, there are two cases for Match: one for the case
where we have exhausted the list of patterns to match a value
against, and one for the case where there are more cases
to consider. In case we have exhausted the list of patterns
to match a value against, we now use mzero to return an
empty set of result threads. Otherwise, we match a value
against a pattern, using the side-effectful vmatchs function
(elided for brevity). If the value contains symbolic variables,

the vmatchs function computes a unifier to be be applied to
the symbolic variables in order to make the pattern match
succeed. The transition function returns the thread resulting
from applying that unifier to the matched branch, unioned
with (via the ‘mplus‘ operation of the SymbolicSetMonad)
any other threads contained in branches with patterns that
may succeed to match (via the recursive call to steps in the
second Match case above). This way, the transition function
computes the set of all possible execution paths for a given
expression.

The case of the steps function above for expressions of the
form Eqc v1 v2 checks whether v1 and v2 are unifiable. If
they are unifiable with the empty unifier, there is only one
possible execution path to consider, namely the execution
path where v1 and v2 are equal. Otherwise, if v1 and v2 have a
non-empty unifier, there are two possible execution paths to
consider: one where v1 and v2 are equal, and one where they
are not. The steps function returns the union (again, using
‘mplus‘) of two threads representing each of these execution
paths. For safety, we register a negative unification constraint
for the execution path that disequates v1 and v2 , such that
v1 and v2 cannot be unified at any point in the future during
symbolic execution.

Driver Loop The driver loop for symbolic execution is gen-
eralized to operate on sets of possible execution paths, where
each execution path is given by a configuration Configs:

type Configs a = (a, Env SymbolicValue,UnifierN )

drives :: [Configs (Threads SymbolicValue)] →
SymbolicMonad (Configs SymbolicValue,

[Configs (Threads SymbolicValue)])
drives [ ] = throwError "No solution found"

drives ts =
case isDone ts of
(Just c, ts′) → return (c, ts′)

→ do
ts′← iterate ts
drives ts′

A configuration comprises a value, an environment which
may contain terms with symbolic variables, and a list of neg-
ative unification constraints (UnifierN ). The drives function
takes a list of configurations as input, uses isDone to check if
one of the input configurations is a value, and returns a pair
of that configuration and the remaining configurations. If
none of the input configurations are values, each input con-
figuration is iterated by a single transition step, and drives is
called recursively on the resulting list of configurations.

AConstraint Language for Symbolic Execution Wehave
shown how to alter the interpretation of the effects in the
definitional interpreter presented in Fig. 2, to derive a sym-
bolic executor from the concrete definitional interpreter from
§ 3. Invoking this symbolic executor with input programs

data Constraint = CTake Int ExConstraint
data ExConstraint = CEx String ExConstraint

| CEq Expr Expr
| CNEq Expr Expr

Figure 3. Syntax for a tiny constraint language

that contain symbolic variables gives rise to a breadth-first
search over possible instantiations of symbolic variables, to
synthesize concrete terms. We provide programmers with
control over which parts of a program (s)he wishes to syn-
thesize by defining a small constraint language on top of the
definitional interpreter from Fig. 2.

The syntax for this constraint language is summarized in
Fig. 3. CTake n cx is a “top-level” constraint for picking n
solutions to a constraint cx that contains existentially quanti-
fied symbolic variables. CEx x cx introduces an existentially
quantified symbolic variable, by populating the environment
of a symbolic interpreter with a symbolic variable value
binding SymV xf for x, where xf is a fresh symbolic variable
name. CEq e1 e2 is a constraint that e1 and e2 evaluate to
the same value, and CNEq e1 e2 is a constraint that e1 and e2
evaluate to different values.
Our approach to constraint solving is given by the solve

function in Fig. 4 which, in turn, calls the searchs function
whose type signature is shown in the figure, but whose imple-
mentation we omit for brevity. searchs e ts ceq n implements
a naive constraint solving strategy which uses a symbolic
executor to search for n different instantiations of symbolic
variables that make the result of symbolic execution of the
input expression e equal to the result of symbolic execu-
tion of a configuration in ts, modulo a custom notion of
SymbolicEquality.

Example: SynthesizingAppendExpressions To illustrate
what we can dowith our derived symbolic executor and small
constraint language, let us consider list concatenation as an
example, inspired by the relational programming techniques
and examples given by Byrd et al. [6]. The append0 program
below grabs a single solution to the constraint which equates
"q" and the result of concatenating (append) a list consisting
of three atoms (a, b, c) with a list of two atoms (d, e):

append0 :: Constraint
append0 =

grab 1 (exists "q"
((append @@ (atom "a" ‘cons‘ (atom "b"

‘cons‘ (atom "c" ‘cons‘ nil)))
@@ (atom "d" ‘cons‘ (atom "e" ‘cons‘ nil)))

‘CEq‘ (var "q")))

Here, append is a recursive function defined in the language
we are symbolically executing (Fig. 1), and@@ is syntactic

solve :: Constraint → SymbolicMonad [Env SymbolicValue ]
solve (CTake n cx) = solvex cx n

solvex :: ExConstraint → Int →
SymbolicMonad [Env SymbolicValue ]

solvex (CEx x cx) n = do
nx ← fresh′

Reader .local (λnv → (x, SymV nx) : nv) (solvex cx n)
solvex (CEq e1 e2) n = do
nv ← ask
searchs e1 [(interp e2, nv, [ ])] unify n

solvex (CNEq e1 e2) n = do
nv ← ask
searchs e1 [(interp e2, nv, [ ])]

(λv1 v2 → case unify v1 v2 of
Just → Nothing
Nothing → Just [ ])

n

type SymbolicEq =
SymbolicValue→ SymbolicValue→ Maybe Unifier

searchs :: Expr →
[Configs (Threads SymbolicValue)] →
SymbolicEq→
Int →
SymbolicMonad [Env SymbolicValue ]

Figure 4. A constraint solver for symbolic execution con-
straints

sugar for ‘App‘. Solving the append0 constraint yields the
instantiation of q to the list contaning all input atoms in
sequence.
We can also use symbolic execution to synthesize inputs

to functions:
append01 :: Constraint
append01 =
grab 1 (exists "q"
((append @@ (var "q")

@@ (atom "d" ‘cons‘ (atom "e" ‘cons‘ nil)))
‘CEq‘ (atom "a" ‘cons‘ (atom "b" ‘cons‘ (atom "c"

‘cons‘ (atom "d" ‘cons‘ (atom "e" ‘cons‘ nil)))))))
Solving the append01 constraint yields the instantiation of q
to the list containing the atoms a, b, c.

We can even use symbolic execution to synthesizemultiple
inputs:
append02 :: Constraint
append02 =
grab 6 (exists "x" (exists "y"
((append @@ (var "x")@@ (var "y"))

17

‘CEq‘ (atom "a" ‘cons‘ (atom "b" ‘cons‘ (atom "c"

‘cons‘ (atom "d" ‘cons‘ (atom "e" ‘cons‘ nil))))))))

Solving the append02 constraint yields the 6 different possi-
ble instantiations of x and y that satisfy the constraint.

5 Correctness
We have shown how to derive a symbolic executor from a
concrete semantics. The derivation was driven by an intu-
itive understanding of what needs to happen in a symbolic
executor (instantiating and refining symbolic variables, fork-
ing new threads of interpretation) in order to ensure that the
symbolic executor explores all possible execution paths, but
only possible execution paths (i.e., no execution paths that do
not correspond to an actual execution path). In this section
we conjecture a correctness proposition for our symbolic
evaluator, and discuss directions for making this correctness
proposition more formal.

Let runStepss be a function that uses the drives function to
drive an expression to a final value and pool of alternative
execution paths that may yet yield a final result:

runStepss :: Expr → Env SymbolicValue→
Either String (SymbolicValue,

[Configs (Threads SymbolicValue)])

We conjecture that, for any pair of concrete environment nv
and symbolic environment nvs that are equal up-to-unification:

1. Any concrete execution path, given by calling runSteps
from § 3 under nv with any e::Expr either yields a value
that is equal up-to-unification to the SymbolicValue
that runStepss returns; or yields a value that one of the
configurations in runStepss will eventually yield, if we
were to iterate that configuration.

2. Any symbolic execution path, given by calling runStepss
under nvs with any e :: Expr yields a symbolic value
and set of configurations that exhaustively describe
any concrete execution path resulting from evaluating
e under any nv ′ that is equal up-to-unification to nvs .

We believe that abstract interpretation [13] is a suitable
framework for formalizing the correspondence between con-
crete and symbolic execution.3 The methodology due to Kei-
del et al. [26] for defining static analyzers with compositional
soundness proofs is attractive to consider for this purpose.
But it is an open question how the small-step interpreta-
tion strategy based on free monads that we adopted in § 3
and § 4 to realize our symbolic executor fits into the frame-
work and methodology of Keidel et al. [26]. In very recent
work, Rozplokhas et al. [36] provide a certified definition of
miniKanren. In future work, we will investigate how to port
their verification technique to the development in this paper.

3Indeed, it seems Cousot [12] has considered how to formalize symbolic ex-
ecution within the framework of abstract interpretation. This formalization
is only available in French [11].

6 Case Study: Automatic Test Generation
for Definitional Interpreters

In order to test the symbolic executor we have developed,
we defined various interpreters for the simply-typed lambda
calculus, and attempt to synthesize program terms that yield
different results for correct and wrong interpreters. Specifi-
cally, we have implemented a canonical, environment-based
interpreter, and variations on this interpreter with scoping
mistakes. Symbolic execution is able to automatically synthe-
size test programs that will detect these mistakes, by looking
for programs whose results differ between the correct inter-
preter and the wrongly-scoped interpreter. For brevity, we
omit discussion of these test cases. The Haskell version of
this paper contains the test cases that we invite interested
readers to consult. Using GHCi (v8.6.4), symbolic execution
takes <1s to synthesize each test program.
Byrd et al. [6] also compare interpreters with lexical and

dynamic scope in their functional pearl on using miniKan-
ren to solve programming problems. Their implementation
is engineered to use miniKanren’s relational programming
constructs to allow them to yield example terms more ef-
ficiently than naively written interpreters. Our case study
does not come near the efficiency of the interpreters with
lexical and dynamic scope of Byrd et al. [6], which synthe-
size 100 example programs in <2s. But we did not attempt to
optimize the interpreter implementations either, neither at
the meta-language nor the object-language level, to make it
easier for the symbolic execution strategy to find solutions.

7 Conclusion
In this paper we studied how to derive a symbolic execu-
tor from concrete definitional interpreters, and presented
techniques for structuring definitional interpreters to ease
this derivation: free monads for compiling a definitional in-
terpreter into a command tree with a small-step execution
strategy, suitable for forking threads of interpretation and
doing breadth-first search over how to instantiate symbolic
variables in ways that correspond to execution paths through
a program, subject to constraints. We introduced a small con-
straint language on top of our symbolic executor, and used
this language to derive test cases for definitional interpreters
for the simply-typed lambda calculus.
In future work, we intend to explore how to make the

derivation techniques presented in this paper formally cor-
rect, how to automate them, and how to make them effi-
ciently executable, akin to, e.g., miniKanren [5, 16].

